Lösung Beispiel 5, Helmut Hörner Mat.Nr. 08850092

Definiere Potentialwall (drei Bereiche 0, v0, 0)

In[1]:= `pot = {0, v0, 0};`

Schrödingergleichung für die drei Bereiche i = 1...3

In[2]:=
$$\text{schrgl} = \text{Table}\left[-\frac{\hbar^2}{2\,m}\,\psi[i]''[x] + \text{pot}[[i]]\,\psi[i][x] == \text{ekin}\,\psi[i][x],\ \{i, 1, 3\}\right]$$
Tabelle

Out[2]=
$$\left\{-\frac{\hbar^2\,\psi[1]''[x]}{2\,m} == \text{ekin}\,\psi[1][x],\ \text{v0}\,\psi[2][x] - \frac{\hbar^2\,\psi[2]''[x]}{2\,m} == \text{ekin}\,\psi[2][x],\ -\frac{\hbar^2\,\psi[3]''[x]}{2\,m} == \text{ekin}\,\psi[3][x]\right\}$$

Allgemeine Lösung der Schrödingergleichung in drei Bereichen i = 1...3

In[3]:=
$$\text{rules1} = \{m > 0,\ \text{ekin} > 0,\ \hbar > 0,\ \text{v0} > 0,\ \text{ekin} < \text{v0}\};$$

$$\text{substEnergy} = \left\{\frac{\sqrt{2\,(\text{v0}\,m - \text{ekin}\,m)}}{\hbar} \to \alpha,\ \frac{\sqrt{2\,\text{ekin}\,m}}{\hbar} \to k\right\};$$

$$\text{substAlphaK} = \left\{\alpha \to \frac{\sqrt{2\,(\text{v0}\,m - \text{ekin}\,m)}}{\hbar},\ k \to \frac{\sqrt{2\,\text{ekin}\,m}}{\hbar}\right\};$$

```
lsgAllg = Table[DSolve[schrgl[[i]], ψ[i][x], x, Assumptions → rules1], {i, 1, 3}] /.
    substEnergy // Flatten
```
Tabelle | löse Differentialgleichung | Annahmen | ebne ein

Out[6]=
$$\{\psi[1][x] \to C[1]\,\text{Cos}[k\,x] + C[2]\,\text{Sin}[k\,x],\ \psi[2][x] \to e^{x\,\alpha}\,C[1] + e^{-x\,\alpha}\,C[2],\ \psi[3][x] \to C[1]\,\text{Cos}[k\,x] + C[2]\,\text{Sin}[k\,x]\}$$

Ersetzen der wiederkehrenden Parameter C[1] und C[2] durch individuelle Parameter

In[7]:=
```
parameter = {{a1, b1}, {a2, b2}, {a3, b3}};
lsg2 =
  Table[lsgAllg[[i]] /. {C[1] → parameter[[i, 1]], C[2] → parameter[[i, 2]]}, {i, 1, 3}]
```
Tabelle | Konstante | Konstante

Out[8]=
$$\{\psi[1][x] \to \text{a1}\,\text{Cos}[k\,x] + \text{b1}\,\text{Sin}[k\,x],\ \psi[2][x] \to \text{b2}\,e^{-x\,\alpha} + \text{a2}\,e^{x\,\alpha},\ \psi[3][x] \to \text{a3}\,\text{Cos}[k\,x] + \text{b3}\,\text{Sin}[k\,x]\}$$

Darstellung in der Form $e^{i k x}$

In[9]:=
```
lsg3 = Collect[TrigToExp[lsg2], ⅇ^_] /. {a_/2 - ⅈ b_/2 → a, a_/2 + ⅈ b_/2 → b}
```
gruppier··· konvertiere trigonometrische Funktion in Exponentialfunktion

Out[9]= $\{\psi[1][x] \to b1\, e^{-i k x} + a1\, e^{i k x},\ \psi[2][x] \to b2\, e^{-x \alpha} + a2\, e^{x \alpha},\ \psi[3][x] \to b3\, e^{-i k x} + a3\, e^{i k x}\}$

Ann: Welle läuft von links nach rechts -> keine Reflexion im Bereich 3

In[10]:=
```
lsg4 = lsg3 /. b3 → 0
```

Out[10]= $\{\psi[1][x] \to b1\, e^{-i k x} + a1\, e^{i k x},\ \psi[2][x] \to b2\, e^{-x \alpha} + a2\, e^{x \alpha},\ \psi[3][x] \to a3\, e^{i k x}\}$

Anschlussbedingungen

In[11]:=
```
stetigkeit = {(lsg4[[1, 2]] /. x → 0) == (lsg4[[2, 2]] /. x → 0),
   (lsg4[[2, 2]] /. x → x1) == (lsg4[[3, 2]] /. x → x1)};
stetigkeitAbl = {(D[lsg4[[1, 2]], x] /. x → 0) == (D[lsg4[[2, 2]], x] /. x → 0),
   (D[lsg4[[2, 2]], x] /. x → x1) == (D[lsg4[[3, 2]], x] /. x → x1)};
anschlussbed = Join[stetigkeit, stetigkeitAbl]
```
leite ab · leite ab · leite ab · leite ab · verknüpfe

Out[13]= $\{a1 + b1 == a2 + b2,\ b2\, e^{-x1 \alpha} + a2\, e^{x1 \alpha} == a3\, e^{i k x1},\ i\, a1\, k - i\, b1\, k == a2\, \alpha - b2\, \alpha,\ -b2\, e^{-x1 \alpha}\, \alpha + a2\, e^{x1 \alpha}\, \alpha == i\, a3\, e^{i k x1}\, k\}$

Berechnung der Parameter aus den Anschlussbedingungen

In[14]:=
```
parameterSol = FullSimplify[Solve[anschlussbed, {b1, a2, b2, a3}]] // Flatten
```
vereinfache voll··· löse · ebne ein

Out[14]=
$$\left\{b1 \to \frac{a1\,(k^2+\alpha^2)\,\mathrm{Sinh}[x1\,\alpha]}{2\,i\,k\,\alpha\,\mathrm{Cosh}[x1\,\alpha] + (k-\alpha)\,(k+\alpha)\,\mathrm{Sinh}[x1\,\alpha]},\ a2 \to -\frac{2\,a1\,k\,(k - i\,\alpha)}{-(k - i\,\alpha)^2 + e^{2 x1 \alpha}\,(k + i\,\alpha)^2},\right.$$
$$\left. b2 \to \frac{2\,a1\,e^{2 x1 \alpha}\,k\,(k + i\,\alpha)}{-(k - i\,\alpha)^2 + e^{2 x1 \alpha}\,(k + i\,\alpha)^2},\ a3 \to \frac{2\,i\,a1\,e^{-i k x1}\,k\,\alpha}{2\,i\,k\,\alpha\,\mathrm{Cosh}[x1\,\alpha] + (k-\alpha)\,(k+\alpha)\,\mathrm{Sinh}[x1\,\alpha]}\right\}$$

Fertige Wellenfunktion für alle drei Bereiche

In[15]:=
```
regeln = {Im[m] == 0, Im[k] == 0, Im[α] == 0, Im[x1] == 0, Im[ℏ] == 0,
         Imaginärteil  Imaginärteil  Imaginärteil  Imaginärteil  Imaginärteil
   Im[v0] == 0, Im[ekin] == 0, ekin < v0, m > 0, ℏ > 0, ekin > 0, v0 > 0};
          Imaginärteil
lsgFinal = FullSimplify[lsg4 /. parameterSol, Assumptions → regeln]
           vereinfache vollständig               Annahmen
psiPiecewiese[x_, A_, x1_] = Piecewise[
                             stückweise
   {{lsgFinal[[1, 2]], x < 0}, {lsgFinal[[3, 2]], x > x1}}, lsgFinal[[2, 2]]] /. {a1 → A}
```

Out[16]=
$$\left\{\psi[1][x] \to a1\, e^{-i k x}\left(e^{2 i k x} + \frac{(k^2+\alpha^2)\,\mathrm{Sinh}[x1\,\alpha]}{2\, i\, k\,\alpha\,\mathrm{Cosh}[x1\,\alpha] + (k-\alpha)\,(k+\alpha)\,\mathrm{Sinh}[x1\,\alpha]}\right),\right.$$
$$\psi[2][x] \to -\frac{2\, a1\, k\,\left(-i\,\alpha\,\mathrm{Cosh}[(x-x1)\,\alpha] + k\,\mathrm{Sinh}[(x-x1)\,\alpha]\right)}{2\, i\, k\,\alpha\,\mathrm{Cosh}[x1\,\alpha] + (k-\alpha)\,(k+\alpha)\,\mathrm{Sinh}[x1\,\alpha]},$$
$$\left.\psi[3][x] \to \frac{2\, i\, a1\, e^{i k (x-x1)}\, k\,\alpha}{2\, i\, k\,\alpha\,\mathrm{Cosh}[x1\,\alpha] + (k-\alpha)\,(k+\alpha)\,\mathrm{Sinh}[x1\,\alpha]}\right\}$$

Out[17]=
$$\begin{cases} A\, e^{-i k x}\left(e^{2 i k x} + \frac{(k^2+\alpha^2)\,\mathrm{Sinh}[x1\,\alpha]}{2\, i\, k\,\alpha\,\mathrm{Cosh}[x1\,\alpha]+(k-\alpha)\,(k+\alpha)\,\mathrm{Sinh}[x1\,\alpha]}\right) & x < 0 \\ \frac{2\, i\, A\, e^{i k (x-x1)}\, k\,\alpha}{2\, i\, k\,\alpha\,\mathrm{Cosh}[x1\,\alpha]+(k-\alpha)\,(k+\alpha)\,\mathrm{Sinh}[x1\,\alpha]} & x > x1 \\ -\frac{2\, A\, k\,(-i\,\alpha\,\mathrm{Cosh}[(x-x1)\,\alpha]+k\,\mathrm{Sinh}[(x-x1)\,\alpha])}{2\, i\, k\,\alpha\,\mathrm{Cosh}[x1\,\alpha]+(k-\alpha)\,(k+\alpha)\,\mathrm{Sinh}[x1\,\alpha]} & \mathrm{True} \end{cases}$$

Transmissionskoeffizient (einmal abh. von k und α; und einmal abh. von v0, ekin und x1)

In[18]:=
```
T1 = ComplexExpand[FullSimplify[Abs[a3/a1 /. parameterSol], Assumptions → regeln]]^2
T2 = FullSimplify[
     vereinfache vollständig
   FullSimplify[T1 /. Cosh[u_]^2 → 1 + Sinh[u]^2] /. substAlphaK , Assumptions → regeln]
   vereinfache vollständig                                       Annahmen
```

Out[18]=
$$\frac{4\, k^2\, \alpha^2}{4\, k^2\, \alpha^2\, \mathrm{Cosh}[x1\,\alpha]^2 + (k-\alpha)^2\, (k+\alpha)^2\, \mathrm{Sinh}[x1\,\alpha]^2}$$

Out[19]=
$$\frac{1}{1 + \frac{v0^2\, \mathrm{Sinh}\left[\frac{\sqrt{2}\,\sqrt{m\,(-ekin+v0)}\; x1}{\hbar}\right]^2}{-4\, ekin^2 + 4\, ekin\, v0}}$$

Reflexionskoeffizient (einmal abh. von k und α; und einmal abh. von v0, ekin und x1)

In[20]:=
```
R1 = ComplexExpand[FullSimplify[Abs[b1/a1 /. parameterSol], Assumptions → regeln]]^2
R2 = Together[FullSimplify[R1 /. substAlphaK , Assumptions → regeln]]
```
zusammen vereinfache vollständig Annahmen

Out[20]=
$$\left(\frac{k^2\sqrt{\text{Sinh}[x1\,\alpha]^2}}{\sqrt{4\,k^2\,\alpha^2\,\text{Cosh}[x1\,\alpha]^2+(k-\alpha)^2\,(k+\alpha)^2\,\text{Sinh}[x1\,\alpha]^2}}+\frac{\alpha^2\sqrt{\text{Sinh}[x1\,\alpha]^2}}{\sqrt{4\,k^2\,\alpha^2\,\text{Cosh}[x1\,\alpha]^2+(k-\alpha)^2\,(k+\alpha)^2\,\text{Sinh}[x1\,\alpha]^2}}\right)^2$$

Out[21]=
$$\frac{v0^2-v0^2\,\text{Cosh}\left[\frac{2\sqrt{2}\sqrt{m\,(-ekin+v0)}\;x1}{\hbar}\right]}{8\,ekin^2-8\,ekin\,v0+v0^2-v0^2\,\text{Cosh}\left[\frac{2\sqrt{2}\sqrt{m\,(-ekin+v0)}\;x1}{\hbar}\right]}$$

Probe: R+T=1

In[22]:=
```
FullSimplify[{R1 + T1, R2 + T2}]
```
vereinfache vollständig

Out[22]= {1, 1}

MANIPULATE: Realteil der Wellenfunktion über alle drei Bereiche

In[23]:=
```
physconst = {ℏ → 6.62606896 × 10^-34, m → 9.10938215 × 10^-31, echrg → 1.60217649 × 10^-19};
Manipulate[Show[
  Plot[Re[psiPiecewiese[xx, 1 / 2, x1] /. substAlphaK /. {ekin → Min[EKIN, V0] echrg,
       v0 → V0 echrg} /. physconst], {xx, -6 × 10^-9, 6 × 10^-9}, PlotRange → {-1, 1}],
  Graphics[{Thick, Red, Line[{{0, -2}, {0, 2}}], Line[{{x1, -2}, {x1, 2}}]}]],
 {{V0, 11}, 0.02, 11}, {{EKIN, 10}, 0, 11 - 0.01},
 {{x1, 2 × 10^-9}, 0 × 10^-9, 3 × 10^-9}]
```
manipuliere | zeige an | stell··· | Realteil | kleinstes Element | Koordinatenbereich der Graphik | Graphik | dick | rot | Linie | Linie

Out[24]=



MANIPULATE: Aufenthaltswahrscheinlichkeit über alle drei Bereiche

In[25]:=
```
Manipulate[Show[
  Plot[Abs[psiPiecewiese[xx, 1 / 2, x1] /. substAlphaK /. {ekin → Min[EKIN, V0] echrg,
        v0 → V0 echrg} /. physconst]^2, {xx, -6 × 10^-9, 6 × 10^-9}, PlotRange → {0, 1}],
  Graphics[{Thick, Red, Line[{{0, -2}, {0, 2}}], Line[{{x1, -2}, {x1, 2}}]}]],
 {{V0, 11}, 0.02, 11}, {{EKIN, 10}, 0, 11 - 0.01},
 {{x1, 0.5 × 10^-9}, 0 × 10^-9, 3 × 10^-9}]
```
manipuliere · zeige an · stelle Funktion graphisch dar · Koordinatenbereich der Graphik · Graphik · dick · rot · Linie · Linie

Out[25]=